
Spring 2018

The example code here is written for Particle Photon to send data (for example,  from sensors) to IBM
Cloud/Watson  IoT Platform,  receive  commands  from IBM Cloud,  parse  received  commands,  and  take
certain actions according to the commands. On the cloud side, a Node-RED application is running to receive
sensor data, analyze data, and generate certain commands according the data. You can create your own
Node-RED application allowing users to initiate a command (for example, a text/Twitter message). 

The MQTT library used in this code is accessible from online Particle Build, which was written by Hirotaka
Niisato [4].  Due to possible update of the MQTT library, the code here might need some modifications.
Search and read  MQTT.h and MQTT.cpp on Particle Build or Niisato’s GitHub repository[4] for complete
detail of the latest MQTT library.  

Thoroughly read the following code, and revise it for your needs.  Note:  To help you understand the code,
some of the important comments are highlighted. 

/*This code is to send an integer dummydata (changing from 0 to 100) to IBM Watson IoT
Platform, subscribe commands from a Node-RED app running on IBM Cloud, and blinks
corresponding indicator LEDs according to the received commands.
 
Note: the Node-RED app on IBM Cloud sends command “turnlow” back to Photon if the app
receives a dummydata in 0~32, sends command “turnmid” if receiving a daummydata in
33~66, “turnhigh” if receiving a dummydata in 67~100. 
*/

/* Author: Xiaohai Li
 * License: GPL v2.0 
 * 05/05/2018
 */

// This #include statement was automatically added by the Particle IDE.
#include <MQTT.h>        //search and include the mqtt library via Particle online IDE

// This #include statement was automatically added by the Particle IDE.
#include <RdJson.h>

#define HOST_PORT  1883 //port# used by host (IBM Watson IoT Platform): 1883 (default)
 //change part# to 443 for secure connection

#define MQTT_QoS   0 //MQTT QoS = 0: message will be delivered zero or once (default)
//MQTT QoS = 1: message will be delivered at least once
//MQTT QoS = 2: message will be delivered exactly once

//Use your device information to fill in the following code in line20~24
char *MQTT_HOST = "your_organizationID_here.messaging.internetofthings.ibmcloud.com";  
char *MQTT_CLIENT = "d:your_organizationID_here:your_Device_Type_here:your_DeviceID_here";  
                   //an example of MQTT_CLIENT is "d:myorgID:photons:photon_a1";
char *MQTT_USERNAME = "use-token-auth";
char *MQTT_PASSWORD = "your_token_here"; 

// Set MQTT event topic for the data that will be published to Watson IoT Platform.
// iot-2    --> The protocol
// evt      --> Specifies the message type, use "cmd" for applications
// testevent--> Name of event        //it is also EventID. Important! Needed by cloud!
// fmt/json --> Message will be send in JSON format //Watson Platform uses JSON format as default
char *EVENT_TOPIC = "iot-2/evt/testevent/fmt/json";     // Segment can be customized by you  

//Specify topic of the commands that Photon will subscribe (which is sent from IBM Cloud).
//Set CommandID as “testcommand”, which should match CommandType setting in Node-RED app
//char *COMMAND_TOPIC = "iot-2/cmd/testcommand/fmt/json"   //use json formatted commands
char *COMMAND_TOPIC = "iot-2/cmd/testcommand/fmt/string";  //use string formatted commands

MQTT client( MQTT_HOST, HOST_PORT, callback ); //Create a MQTT client with callback function

char payload[80];     //MQTT data message's payload. A string in JASON format.  
                      // 80Bytes used here. Change the size according to your data.

int dummydata = 0;

int LedIndicator_Publish = D7;   //set LED D7 as publishing indicator
int LedIndicator_CommLow = D6;   //set LED connected on D6 as indicator for command “turnlow”
int LedIndicator_CommMid = D5;   //set LED connected on D6 as indicator for command “turnmid”
int LedIndicator_CommHigh = D4;  //set LED connected on D6 as indicator for command “turnhigh”

void BlinkLed(int LedPin, int BlinkTimes, int BlinkPeriod);  //A func to blink a LED

void setup() {    
  pinMode(LedIndicator_Publish, OUTPUT);
  pinMode(LedIndicator_CommLow, OUTPUT);
  pinMode(LedIndicator_CommMid, OUTPUT);
  pinMode(LedIndicator_CommHigh, OUTPUT);
  
  BlinkLed(LedIndicator_CommLow, 1, 300);
  BlinkLed(LedIndicator_CommMid, 1, 300);
  BlinkLed(LedIndicator_CommHigh, 1, 300);

  RGB.control(true);

  Serial.begin( 9600 );    //Use serial monitor to debug the code. 
                           //Read “Particle serial tutorial” for detailed how-to. 
  Serial.println( "Connecting Photon to IBM Watson IoT Platform ...... " );

  while( !Serial.available() ) {
    Particle.process();
  }   

  client.connect(MQTT_CLIENT,MQTT_USERNAME,MQTT_PASSWORD);//Connect Photon to Watson IoT Platform

  if( client.isConnected() ) {              //Verify the connection
    Serial.println( "Now connected!" );
    client.subscribe(COMMAND_TOPIC);    //subscribe commands with CommandType specified in line36
  }
}

void loop() {
  dummydata ++;
  if (dummydata > 100)  dummydata = 0;

  //Convert data into a JSON formatted object by line86:
  //JSON format example: {“property”: value1, ”property2”: value2,…}
  sprintf( payload, "{ \"dataproperty\": \"%d\" }", dummydata ); 
  
  //Publish the JSON formatted payload to IBM cloud under the pre-defined Event Topic in line31
  client.publish( EVENT_TOPIC, const_cast<char*> (payload) ); 
  
  BlinkLed(LedIndicator_Publish, 1, 200); //blink the publishing LED once when a data is sent
      
  client.loop();

  Serial.print( "Data being sent to Cloud: " ); //Display the sent data on SerialMonitor 
  Serial.println( dummydata );

  delay( 5000 );  //delay 5sec: Set the period of publishing data to cloud
}

void callback( char* topic, byte* payload, unsigned int length ) {  
  RGB.color(255,10,255);  //blink the RGB LED in pink color
  delay(300); 
  RGB.color(0,0,0);
  
  char p[length + 1];
    
  memcpy( p, payload, length );  //read in received command message payload
  p[length] = NULL;

  String message( p );

  Serial.print( "Received command from Cloud: ");  //display received command on serial terminal
  Serial.println(p);

 //Line118: to parse received json object to obtain commands. If the commands are in string 
 //format, no need of this. 
 // String CmdStr = RdJson::getString("testcommand", "", p.c_str()); 
 // Serial.printlnf("Received Command from cloud: %s", CmdStr.c_str());

 //Line122-138: Take diff actions according to received commands. Make changes for your needs.
  if (!strcmp(p, "turnlow"))   
  {
      BlinkLed(LedIndicator_CommLow, 1, 300);
  }  
  else if (!strcmp(p, "turnmid")) 
  {
      BlinkLed(LedIndicator_CommMid, 1, 300);
  }
  else if (!strcmp(p, "turnhigh") )          
  {
      BlinkLed(LedIndicator_CommHigh, 1, 300);
  }  
  else {
      RGB.color(255,255,255); delay(1000); RGB.color(0,0,0);  //Blink the RGB LED in white
      BlinkLed(LedIndicator_CommLow, 1, 100);
      BlinkLed(LedIndicator_CommMid, 1, 100);
      BlinkLed(LedIndicator_CommHigh, 1, 100);
  }
}

void BlinkLed(int LedPin, int BlinkTimes, int BlinkPeriod) //A func to blink a LED for BlinkTimes
{
    for (int k=0; k<BlinkTimes; k++){
        digitalWrite(LedPin, HIGH);
        delay(BlinkPeriod);
        digitalWrite(LedPin, LOW);
        delay(BlinkPeriod);
    }    
}

Note:   Refer to Particle Serial Tutorial to learn how to use Serial Monitor with Particle for testing and
debugging. 
Link:  https://community.particle.io/t/serial-tutorial/26946
or: https://github.com/rickkas7/serial_tutorial

